
Enhancing Database Security:

Monitoring Audit Trails Using Enterprise Manager

Peter J. Magee, CDA Gail VanderKolk
SQRIBE Technologies Reynolds & Reynolds

Abstract
Maintaining the security and integrity of a corporation’s sensitive data is of great importance to the database
administrator, particularly in today’s intra- and inter-networked environments.  This presentation introduces a
technique - developed initially for the Department of Defense - which enables real time notification of audited
events through the Oracle Enterprise Manager Console.  Scripts and utilities will be presented that allow the DBA
to gain more information from the audit trail than that provided by the basic DBA views, and to use that
information to identify more rapidly and accurately any unauthorized activity or system intruders.  Use of these
tools in Oracle7 or Oracle8 databases to consolidate security monitoring in a distributed environment will be
illustrated.

Introduction
Operating systems generally record security events such as logins, valid or invalid, in an audit trail.  Most systems
are configured to automatically suspend an account after several invalid login attempts, and many can be set to
notify the system administrator in real-time of errors via SNMP monitoring.  Other events can also be monitored,
such as unauthorized file accesses and deletions, service and port usage, and program execution.

Similar to an operating system, Oracle can also be configured to audit a wide variety of activities and events,
including user sessions, object creation and modification, privilege and resource usage, and even (using triggers)
data field value changes.  Unfortunately, Oracle provides no default mechanism for automatically notifying the
DBA when any of these audited events take place, and no easy way to obtain some of the most valuable information
in the audit trail: the identity and network location of the offending user.  Beginning with version 7.3 Oracle does,
however, provide all the tools necessary to eliminate this shortcoming.

Oracle Enterprise Manager (OEM) allows the DBA to monitor a number of specially provided events in the
database via SQL*Net and SNMP.  Although current versions of OEM do not allow database administrators to
create their own customized events, it is possible to take advantage of one of the packaged OEM events and the
PL/SQL package UTL_FILE to allow real-time monitoring of security violations.  The Alert event, packaged with
Oracle’s Diagnostic Pack, scans the alert log of a specified database for any new error messages.  When an Oracle
“ORA” message like ORA-00600 (internal error) or ORA-07445 (core dump) is found, the error number and
severity are sent to the OEM console where an event is recorded, and the DBA is alerted.  By using a specially
designed audit trail trigger called WatchDog, administrators in Oracle version 7.3 and up can add their own error
messages to the alert log and thus generate custom alerts to the OEM console based on audited events.

Oracle’s Audit Trail
By default, audit trail information is recorded in the AUD$ table, owned by user SYS.  Oracle maintains a list of
events that may be monitored, and the DBA may enable auditing for any or all of them.  When a selected event

occurs, it is registered in the AUD$ table.  The DBA is provided a set of views based on the AUD$ table, including
DBA_AUDIT_OBJECT, DBA_AUDIT_PRIVS, DBA_AUDIT_SESSION, and DBA_AUDIT_TRAIL.  These
views provide some basic reporting functionality on different types of audited events.  The views can be queried
periodically by the DBA to identify security violations.  Unfortunately, manually scanning the audit trails is
extremely time-consuming.  In a networked environment with multiple instances or extremely busy systems, it can
be difficult to be sure that this periodic monitoring will happen at just the right time to respond promptly to a
security breach.  To enable immediate feedback to the DBA when security violations occur, triggers may be used.
Some modifications to the default audit trail are necessary in order to configure Oracle to respond automatically to
audited events.

Although Oracle has established SYS as the owner of the AUD$ table by default, there are some compelling
reasons to modify the audit trail configuration.  Because of its nature, the audit trail will continue to grow
unchecked as long as the database is in use.  In order to prevent the audit trail from taking over the database, it is
necessary to periodically “weed out” or truncate those audit trail records that are no longer of interest.  Since all of
SYS’s objects, including the AUD$ table, are contained in the SYSTEM tablespace, this weeding-out will
eventually cause fragmentation, degrading database performance over time.  The only way to eliminate
fragmentation in the SYSTEM tablespace, once it has occurred, is to recreate the instance.  Therefore, relocating
the AUD$ table to another tablespace allows the DBA to prevent unnecessary fragmentation of the SYSTEM
tablespace.  An ancillary benefit is better control over growth of the audit trail by defining appropriate storage and
growth parameters when the table is created.

To configure Oracle to respond automatically to an audited event, it is necessary to create triggers that will listen
for specific entries into the AUD$ table.  However, Oracle does not permit triggers on any objects owned by SYS
(produces Oracle error ORA-04089: “cannot create triggers on objects owned by SYS”).  In order to monitor
activity in the AUD$ table, then, it must be owned by a user other than SYS.  The solution is to change ownership
of the audit trail table to SYSTEM, and create a view called AUD$ in the SYS schema that points to the new table.

The remainder of this paper discusses a strategy for customizing the audit trail as described above and creating the
WatchDog trigger to identify and log any attempted security breaches.

Modifying the Audit Trail
First, ownership of the AUD$ table must be changed, and it must be relocated out of the SYSTEM tablespace to a
separate auditing tablespace.  An auditing tablespace will be created, then the existing AUD$ table will be renamed
and copied to the SYSTEM schema.  A view named AUD$ will be created in the SYS schema, which points to
SYSTEM.AUD$.  Finally, the auditing portion of the data dictionary will be recreated using the cataudit.sql script.

Special Note
Changing either the location or the ownership of AUD$ is a configuration change that is not supported by
Oracle Technical Support.  Take care when performing the following steps to ensure the procedures are followed
exactly.  While the authors have tested these scripts on a variety of Oracle versions and OS platforms, no
warranty, expressed or implicit, is given that you will have the same results.

Step 1
If auditing is already enabled in the database, it will need to temporarily be disabled while these changes are made.
In the init.ora file, make sure that the audit trail parameter is turned off.  Make sure that the value of this parameter
is set to “none”.  If the value of this parameter is changed, the database will need to be restarted for the change to
take effect.

Step 2
Check to see if there is already an auditing tablespace in the database.  Connect internal using Server Manager and
run the following command.

select tablespace_name from dba_tablespaces;

TABLESPACE_NAME

---------------

SYSTEM

USER_DATA

ROLLBACK_DATA

TEMPORARY_DATA

Step 3
If an auditing tablespace does not exist, execute the following commands as SYS (or internal).  Substitute a file
name appropriate to the local operating system, and enter storage parameters consistent with the rest of the
database.

create tablespace "DBAUDIT"

  datafile '<use appropriate file name>' size 1m

  default storage (initial 128k next 128k pctincrease 0);

Repeat the query from Step 2 to verify that the auditing tablespace has been correctly created.

Step 4
Create the new audit trail table by executing the following commands as SYS (or internal).  Substitute storage
parameters consistent with the rest of the database.

rename aud$ to aud$_temp;

create table system.aud$ tablespace "DBAUDIT" storage (initial 64k next 64k pctincrease 0)

  as select * from aud$_temp;

Step 5
 Connect as SYSTEM and execute the following commands to index and set permissions on the new audit trail
(again substituting appropriate storage parameters):

create index i_aud1 on aud$(sessionid,ses$tid)

  tablespace "DBAUDIT" storage (initial 64k next 64k pctincrease 0);

grant all on aud$ to sys with grant option;

Step 6
Connect internal from Server Manager and run these commands to reset and rebuild the auditing portions of
Oracle’s data dictionary:

create view aud$ as select * from system.aud$;

@$ORACLE_HOME/rdbms/admin/cataudit.sql

(if you are using Oracle for Windows95 or WindowsNT, then run
%ORACLE_HOME%\rdbms73\admin\cataudit.sql or %ORACLE_HOME%\rdbms80\admin\cataudit.sql)

Step 7
SYSTEM now owns the audit trail.  Stop the database and set the following initialization parameters, then restart:

audit_trail = db

utl_file_dir = /your/alert/log/directory/path

where the directory in utl_file_dir is the directory which contains the database alert log; this will allow the audit
trail trigger to update the alert log when an audited event is detected.  Any previously existing audits should be
unaffected; all new entries will be written to the new table owned by SYSTEM.  The WatchDog trigger can now be
written against AUD$ to automate responses to various events that may show up, like failed logins or other unusual
activities.

Step 8
Make sure any audits that need to be monitored are in place.  The following examples will use the session audit to
determine when a login to the database has failed.  The session audit is enabled by a DBA with the following
command, issued from Server Manager.

AUDIT SESSION;

Securing the UTL_FILE Package
Before the WatchDog trigger is created, there are some additional security enhancements that can be made to the
database.  The UTL_FILE package available in Oracle 7.3 and higher allows stored procedures to have access to
operating system files and directories on the server machine.  This package will allow the WatchDog audit trail
trigger to access the database alert log.

UTL_FILE access to operating system directories is controlled by the UTL_FILE_DIR parameter in the init.ora
file for the database.  Only files in designated directories can be accessed by UTL_FILE.  By default, execute on
UTL_FILE is granted to public, which means that any user in the database has potential read-write access as the
Oracle software owner to the UTL_FILE directories (including the alert log, when the WatchDog configuration
changes in Step 7 are made).  This is true even if the user has no other privileges or no user account on the server
operating system.  Execute on UTL_FILE should only be granted on an individual basis to trusted and approved
users (preferably only DBAs or trusted software administrators).

Step 9
This security risk is eliminated by revoking the execute grant from public with the following command, executed as
SYS from Server Manager:

REVOKE EXECUTE ON UTL_FILE FROM PUBLIC;

Step 10
Next, regrant the privilege explicitly to SYSTEM.

GRANT EXECUTE ON UTL_FILE TO SYSTEM;

Expanding the Audit Trail Views
Some additional WatchDog views will enhance the quality of information available to the database administrator.
These views make use of information stored in the AUD$ table, but not broken out in the default DBA_AUDIT
views.

WD_AUDIT_CLIENT_SESSION adds two extra columns to the standard DBA_AUDIT_SESSION view:
COMMENT_TEXT, which contains information about the client connection passed on from the SQL*Net listener,
and IP_ADDRESS, which is stripped from COMMENT_TEXT for TCP protocol connections.
WD_AUDIT_FAILED_SESSION is a subset of WD_AUDIT_CLIENT_SESSION that displays information only
on failed access attempts.  This information is very useful when tracking the source of invalid access attempts.

Special Note
If you are using a connection protocol other than TCP, you may need to modify the WatchDog views below to
successfully extract the network address of the database client.

Step 11
To create these views, connect internal to the database using Server Manager, then run the following scripts:

create or replace view wd_audit_client_session as

select os_username, username, userhost, terminal, timestamp, action_name, logoff_time, logoff_lread,

       logoff_pread, logoff_lwrite, logoff_dlock, sessionid, returncode, comment_text, session_label,

       substr(substr(comment_text,instr(comment_text,'HOST=',1)+5,instr(comment_text,')',

         instr(comment_text,'HOST=',1)) - (instr(comment_text,'HOST=',1)+5)),1,15) ip_address

from dba_audit_trail

where action between 100 and 102;

create or replace view wd_audit_failed_session as

select os_username, username, userhost, terminal, timestamp, action_name,

       sessionid, returncode, comment_text,

       substr(substr(comment_text,instr(comment_text,'HOST=',1)+5,instr(comment_text,')',

         instr(comment_text,'HOST=',1)) - (instr(comment_text,'HOST=',1)+5)),1,15) ip_address

from dba_audit_trail

where returncode in (1017,1005,28000) and action between 100 and 102;

create public synonym wd_audit_client_session for sys.wd_audit_client_session;

create public synonym wd_audit_failed_session for sys.wd_audit_failed_session;

grant select on wd_audit_client_session to dba;

grant select on wd_audit_failed_session to dba;

Creating the Trigger
Sample versions of the WatchDog trigger are shown below.  These WatchDog triggers report access violations that
return codes 1017 or 1005 during a login attempt (invalid userid/password, or null password).  To use these
WatchDogs, insert an appropriate alert file name and directory path where indicated.  WatchDog triggers can be
easily modified to record other audited events as well.

Step 12
As SYS, modify and run the appropriate version of the trigger script to create the WatchDog trigger on
system.aud$.  Replace the <name of log file> and <path to log file> markers with appropriate values like
‘alertSID.ora’ and ‘/your/alert/log/directory/path’.

Oracle7 Version of WatchDog

CREATE OR REPLACE TRIGGER WATCHDOG

AFTER INSERT ON SYSTEM.AUD$

FOR EACH ROW

DECLARE

  FILE_HANDLE  UTL_FILE.FILE_TYPE;

  ALERT_FILE   VARCHAR2(20);

  ALERT_PATH   VARCHAR2(50);

  CURRENT_DATE VARCHAR2(30);

  IP_ADDRESS   VARCHAR2(15);

  PROTOCOL     VARCHAR2(10);

BEGIN

/* Check if latest connection is invalid

   Code 1005 = Null password, 1017 = Invalid userid/password

   Event 100 = LOGON, 101 = LOGOFF, 102 = LOGOFF BY CLEANUP */

  IF (:NEW.RETURNCODE IN (1005, 1017)) AND (:NEW.ACTION IN (100,101,102)) THEN

/* Initialize path and file names */

    ALERT_FILE := '<name of log file>';

    ALERT_PATH := '<path to log file>';

/* Format timestamp for alert log */

    CURRENT_DATE := TO_CHAR(:NEW.TIMESTAMP,'Dy Mon DD HH24:MI:SS YYYY');

/* Determine IP Address of client */

    IP_ADDRESS := SUBSTR(SUBSTR(:NEW.COMMENT$TEXT,INSTR(:NEW.COMMENT$TEXT,'HOST=',1)+5,

      INSTR(:NEW.COMMENT$TEXT,')',INSTR(:NEW.COMMENT$TEXT,'HOST=',1)) –

      (INSTR(:NEW.COMMENT$TEXT,'HOST=',1)+5)),1,15);

/* Determine connection protocol of client */

    PROTOCOL := NVL(SUBSTR(SUBSTR(:NEW.COMMENT$TEXT,INSTR(:NEW.COMMENT$TEXT, 'PROTOCOL=',1)+9,

      INSTR(:NEW.COMMENT$TEXT,')',INSTR(:NEW.COMMENT$TEXT,'PROTOCOL=',1)) -

     (INSTR(:NEW.COMMENT$TEXT,'PROTOCOL=',1)+9)),1,10),'ipc');

/* Open alert log for append */

    FILE_HANDLE := UTL_FILE.FOPEN(ALERT_PATH,ALERT_FILE,'A');

/* Write alert log entry with [DB User][Terminal][OS User][Protocol][IP Address] */

    UTL_FILE.PUT_LINE(FILE_HANDLE,CURRENT_DATE);

    UTL_FILE.PUT_LINE(FILE_HANDLE,'ORA-0'||:NEW.RETURNCODE||': Logon denied ['||:NEW.USERID||

      '] ['||:NEW.TERMINAL||'] ['||:NEW.SPARE1||'] ['||PROTOCOL||'] ['||IP_ADDRESS||']');

/* Close alert log */

    UTL_FILE.FCLOSE (FILE_HANDLE);

  END IF;

EXCEPTION

  WHEN NO_DATA_FOUND THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN UTL_FILE.INVALID_PATH THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN UTL_FILE.READ_ERROR THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN UTL_FILE.WRITE_ERROR THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN OTHERS THEN UTL_FILE.FCLOSE(FILE_HANDLE);

END;

Oracle8 Version of WatchDog

CREATE OR REPLACE TRIGGER WATCHDOG

AFTER INSERT ON SYSTEM.AUD$

FOR EACH ROW

DECLARE

  FILE_HANDLE  UTL_FILE.FILE_TYPE;

  ALERT_FILE   VARCHAR2(20);

  ALERT_PATH   VARCHAR2(50);

  CURRENT_DATE VARCHAR2(30);

  IP_ADDRESS   VARCHAR2(15);

  PROTOCOL     VARCHAR2(10);

BEGIN

  /* Check if connection attempt is invalid

       Code 1005 = Null password, 1017 = Invalid userid/password, 28000 = Account locked (v8 only)

       Event 100 = LOGON, 101 = LOGOFF, 102 = LOGOFF BY CLEANUP */

  IF (:NEW.RETURNCODE IN (1005, 1017, 28000)) AND (:NEW.ACTION# IN (100,101,102)) THEN

/* Initialize path and file names */

    ALERT_FILE := '<name of log file>';

    ALERT_PATH := '<path to log file>';

/* Format timestamp for alert log */

    CURRENT_DATE := TO_CHAR(:NEW.TIMESTAMP#,'Dy Mon DD HH24:MI:SS YYYY');

/* Determine IP Address of client, if available */

    IP_ADDRESS := SUBSTR(SUBSTR(:NEW.COMMENT$TEXT,INSTR(:NEW.COMMENT$TEXT,'HOST=',1)+5,

      INSTR(:NEW.COMMENT$TEXT,')',INSTR(:NEW.COMMENT$TEXT,'HOST=',1)) -

      (INSTR(:NEW.COMMENT$TEXT,'HOST=',1)+5)),1,15);

/* Determine connection protocol of client */

    PROTOCOL := NVL(SUBSTR(SUBSTR(:NEW.COMMENT$TEXT,INSTR(:NEW.COMMENT$TEXT,'PROTOCOL=',1)+9,

      INSTR(:NEW.COMMENT$TEXT,')',INSTR(:NEW.COMMENT$TEXT,'PROTOCOL=',1)) -

      (INSTR(:NEW.COMMENT$TEXT,'PROTOCOL=',1)+9)),1,10),'ipc');

/* Open alert log for append */

    FILE_HANDLE := UTL_FILE.FOPEN(ALERT_PATH,ALERT_FILE,'A');

/* Write alert log entry with [DB User][Terminal][OS User][Protocol][IP Address] */

    UTL_FILE.PUT_LINE (FILE_HANDLE,CURRENT_DATE);

    UTL_FILE.PUT_LINE (FILE_HANDLE,'ORA-0'||:NEW.RETURNCODE||': Logon denied ['||:NEW.USERID||

      '] ['||:NEW.TERMINAL||'] ['||:NEW.SPARE1||'] ['||PROTOCOL||'] ['||IP_ADDRESS||']');

/* Close alert log */

    UTL_FILE.FCLOSE (FILE_HANDLE);

  END IF;

EXCEPTION

  WHEN NO_DATA_FOUND THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN UTL_FILE.INVALID_PATH THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN UTL_FILE.READ_ERROR THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN UTL_FILE.WRITE_ERROR THEN UTL_FILE.FCLOSE(FILE_HANDLE);

  WHEN OTHERS THEN UTL_FILE.FCLOSE(FILE_HANDLE);

END;

Special Note
Be sure to create the WatchDog trigger as SYS. Otherwise you may encounter an ORA-03113 (“end-of-file on
communication channel”) error on the client (i.e. SQL*Plus), and generate an ORA-07445 (“exception
encountered: core dump [%s]”) error or ORA-00600 (“internal message code, arguments: [num]”) error on the
server when you try to log in.

Configuring Enterprise Manager
As mentioned above, the Oracle Enterprise Manager Diagnostics Pack is required to detect events in a database
alert log.  The default event set “Oracle DB Fault” contains the alert event and can be registered with each
database, but this event set also contains the User Blocks and Probe events, which may not be necessary for all
systems.  Instead of using the DB Fault event set, a custom Security event set can be created that contains only
those events that are required.

Step 13
It is assumed that the Oracle Enterprise Manager (OEM) and Diagnostics Pack have been installed, and that the
OEM console is running.  It is also assumed that the DBSNMP services for each monitored database have been
correctly configured and started on each host platform (see the Oracle Enterprise Manager Installation Guide,
Chapters 1 and 2 for complete instructions).  Select the Event Set Library tab in the Event pane. From the Event
menu, select Create Event Set.

Step 14
In the General tab of the Create Event Set dialog box, enter the name of the new event set, “WatchDog Security”.
Make sure that the Service Type field is set to “Database”. Enter a description of the event set, “Custom Database
Security Event Set”.  In the Events tab of the Create Event Set dialog box, double click on the “Alert” event in the
Available Events list.  It should move to the Selected Events list. In the Parameters tab of the Create Event Set
dialog box, set the frequency with which the alert log should be checked for new messages.  A time of several
minutes is generally sufficient.  When the frequency has been correctly set, click the OK button at the bottom of the
dialog box.

Step 15
Select the new event set from the list of event sets in the Event Set Library.  Select Register Event Set from the
Event menu.  In the Register Event Set dialog box, select the databases to be registered from the Available

Destinations.  Each database should appear in the Selected Destinations list as it is selected.  When the desired
databases have been selected, click the OK button at the bottom of the dialog box.

Each selected database should now be registered to report any “ORA” error messages that appear in the alert log
directly to the OEM console.

Summary
Operating systems like UNIX and Windows NT can be configured to keep detailed security logs, and to
automatically notify administrators when an audited event occurs via SNMP or e-mail.  This level of security is no
less important with a data system like Oracle, particularly when users access the database across a network and
never log in to the server operating system.  Being able to accurately identify the source of unauthorized or resource
intensive activity within a system is critical to a DBA’s responsibility for ensuring data integrity and availability.

Using Oracle Enterprise Manager and the Oracle Diagnostic Pack, it is now possible to monitor security-related
events in near real-time, and to easily identify the source and nature of any audited activity.  By configuring a
simple trigger on the database audit trail and making use of the UTL_FILE PL/SQL package, the DBA can write
customized error messages to the Oracle alert log.  The OEM Alert event can be used to scan the alert log for new
error messages and return those messages to the OEM console for the attention of the DBA.  This ability to
automate audit trail monitoring using native Oracle tools dramatically increases the ability of the administrator to
identify security breaches in a timely and efficient manner.



Paper #139 / Page 10

References
Peter J. Magee (1997). Getting More Out Of DBA_AUDIT_SESSION, World Wide Web at
http://members.aol.com/mageedba

Peter J. Magee (1997). How to Place a Trigger on the Audit Trail, World Wide Web at
http://members.aol.com/mageedba

Peter J. Magee (1997). Securing UTL_FILE, World Wide Web at http://members.aol.com/mageedba

Peter J. Magee (1997). WatchDog, World Wide Web at http://members.aol.com/mageedba

Scott Gossett (1996). TFTS: Script to Move SYS.AUD$ Table Out of SYSTEM Tablespace (Problem Number
1019377.6), World Wide Web at http://support.us.oracle.com/ml/plsql/knowledgebase.getcr?textkey=135831

Biography
Peter J. Magee, CDA
SQRIBE Technologies

Mr. Magee worked for over 6 years as a defense contractor for the U.S. Air Force and the Defense Information
Systems Agency (DISA).  For two years he was responsible for Oracle database administration and security at the
DISA Defense Megacenter at Wright-Patterson AFB, where he consulted with numerous military organizations on
the design and implementation of database security protocols.  He has published several security related scripts and
white papers on the world wide web and was a contributing author to the DISA Database Security Technical
Implementation Guide.  Mr. Magee is currently a Senior Systems Administrator and Certified Oracle DBA with
SQRIBE Technologies in Fairborn, Ohio.

Gail W. VanderKolk
Reynolds & Reynolds

Ms. VanderKolk has worked as a defense contractor for the U.S. Air Force for the past 7 years.  During that time,
she has worked on data modeling, logical and physical database design, and as an Oracle Database Administrator.
Ms. VanderKolk is currently a Database Administrator at Reynolds & Reynolds in Dayton, Ohio.


